VI olimpiada

I etapas

58. Kokia skaičiavimo sistema. Paprasčiausias šio uždavinio algoritmo veiksmas – rasti skaičiaus a skaitmenų sumą: tam sudarome atskirą funkciją. 

function sk_suma (sk: integer): integer;

{ randama skaičiaus sk skaitmenų suma }

  var suma: integer;

begin

  suma := 0;

  repeat

    suma := suma + sk mod 10;

    sk := sk div 10

  until sk = 0;

  sk_suma := suma

end; { sk_suma }

Skaitmenų sumą radome dešimtaine sistema. Tuomet skaičių s reikės užrašyti dešimtaine sistema: tam sudarysime atskirą funkciją. Skaičių s iš x sistemos į dešimtainę perrašome šitaip:

__________

s1s2s3...sn  [x] = s1xn-1 + s2xn-2 + s3xn-3 + ... + snx0[10]

čia s1, s2, ... sn – skaičiaus skaitmenys, x – sistemos pagrindas.

function dešimtainis(sk, x: integer): integer;

{ skaičius sk (kurio sistemos pagrindas x) užrašomas dešimtainiu }

  var deš, laip: integer;

begin

  deš := 0;

  laip := 1;   { sistemos pagrindo laipsnių rodikliai }

  repeat

    deš:= deš + sk mod 10 * laip;

    laip := laip * x;

    sk := sk div 10

  until sk = 0;

  dešimtainis := + deš

end; { dešimtainis }

Akivaizdu, kad x > si, (1 ( i ( n). Vadinasi, pirmiausia reikėtų rasti didžiausią duotų skaičių a ir s skaitmenį ir tinkamos sistemos pradėti ieškoti nuo šio skaitmens, padidinto vienetu.

Atkreipiame dėmesį į svarbiausią algoritmo vietą: didinant sistemos pagrindą x, dešimtainis skaičius didėja, pavyzdžiui:

1234 = 27

1235 = 38

1236 = 51

1237 = 66

Vadinasi, pagrindą tereikia keisti tol, kol neviršijama skaičiaus a skaitmenų suma.

type skaitmuo = 0..9;



function did_skaitmuo(sk: integer): skaitmuo;

{ didžiausio skaitmens nustatymas }

  var max: skaitmuo;

begin

  max := 0;

  repeat

    if sk mod 10 > max

       then max := sk mod 10;

    sk := sk div 10

  until sk = 0;

  did_skaitmuo := max

end; { did_skaitmuo }



function sk_sistema (a, s: integer): integer;

{ s > 0, a >= s }

{ randamas skaičiavimo sistemos pagrindas; jei nulis, tai  skaičiavimo sistemos nerasta }

  var m1, m2: skaitmuo;

      x,     { sistemos pagrindas }

      d, suma: integer;

begin { sk_sistema }

  { randame mažiausią galimą skaičiavimo sistemą }

  m1 := did_skaitmuo(s);

  m2 := did_skaitmuo(a);

  if m1 > m2

     then x := m1 + 1

     else x := m2 + 1;

  { randame skaičiaus a skaitmenų sumą }

  suma := sk_suma(a);

  { skaičių s, užrašytą sistema x, paverčiame dešimtainiu }

  d := dešimtainis(s, x);

  if s >= 10

     then while d < suma do

            begin

              x := x + 1;

              d := dešimtainis(s, x)

            end;

  if d = suma

     then sk_sistema := x

     else sk_sistema := 0

end; { sk_sistema }



Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�10212 12�4�610 ( 124��2�320 5�6�510 ( 56��3�5430 9�Sistemos nėra���4�321 14�Sistemos nėra���5�32654 16�14�2010 ( 1614��6�9999 13�33�3610 ( 1333��

59. Parketo klojimas. Įsivaizduokime, kad turime be galo didelį parketo lentelėmis išklotą lauką ir ant jo reikia uždėti stačiakampį (kambarį) taip, kad į jį patektų kuo didesnis sveikų parketo lentelių skaičius (59.2 pav.). 

59.2 pav.

Panagrinėkime, kiek yra skirtingų dėliojimo variantų. Horizontaliai padėta plokštelė gali būti prispausta prie viršutinio kairiojo kampo, gali būti pasislinkusi į dešinę per vieną, du, ... 2(ilg - 1 langelių (59.3.a-c pav.).

59.3.a-c pav. 

Stumdydami stačiakampį aukštyn ar žemyn naujų variantų negausime. Taigi, iš viso turime 2 ( ilg dėliojimo variantų.

Suskaičiuokime, kiek horizontalių plokštelių bus padėta eilutėje, jei pirmoji plokštelė nuo kairiojo krašto pastumta per post1 vietų. Kadangi horizontalios plokštelės dedamos kas ilg langelių, eilutėje tilps (59.4.a-b pav.)

(kilg - post1) div (2ilg)  plokštelių,           jei (kilg - post1) mod (2ilg) < ilg

(kilg - post1) div (2ilg) + 1 plokštelių       priešingu atveju

59.4.a-b pav.

Vienu sakiniu plokštelių skaičių galime užrašyti taip:

(kilg - post1 + ilg) div (2ilg)



Analogiškai galime apskaičiuoti vertikaliai padėtų plokštelių skaičių stulpelyje:

(kpl - post2 + ilg) div (2ilg), 

čia post2 – pirmosios vertikalios plokštelės postūmis nuo viršutinio krašto.

Taigi, kuo pirmoji (horizontali ar vertikali) plokštelė pristumta arčiau krašto, tuo daugiau plokštelių telpa į eilutę (stulpelį). 

Iš formulių matome, kad plokštelių skaičius eilutėse (stulpeliuose) kinta periodiškai. Imkime horizontalias plokšteles ir begalinio pločio kambarį (59.5 pav.).

59.5 pav.



e1 eilučių turi po hor plokštelių, po jų eina e2 eilučių, turinčių hor-1 plokštelių, tuomet vėl e1 eilučių, turinčių po hor-1 plokštelę ir t. t. 

Daugiausia horizontalių plokštelių pateks į kambarį, jei pirmoje eilutėje esanti plokštelė bus prispausta prie krašto. 

Analogiškai yra ir su vertikaliomis plokštelėmis. 

Taigi, jei klodami kambarį galėtume ir horizontalią ir vertikalią plokštelę prispausti prie kairiojo viršutinio kampo, kambaryje padėtų plokštelių skaičius būtų didžiausias. Pavadinkime jį pl.

Tačiau to neįmanoma padaryti. Galimi tik tokie du variantai (59.6.a-b pav.):

59.6.a-b pav.

Pirmuoju atveju horizontalių plokštelių skaičius gali sumažėti vienetu, antruoju – vertikalių. Plokštelių skaičius sumažėja, jei paskutiniojoje eilutėje (stulpelyje) yra hor-1 horizontalių (ar stulpelyje ver-1 vertikalių) plokštelių.

Na o maksimalus plokštelių skaičius kambaryje sumažės, jei plokštelių skaičius sumažėja ir vienu ir kitu atveju 

Įrodėme, kad geriausias išdėstymas gaunamas tada, kai plokštelės prispaudžiamos prie kairiojo viršutinio kampo. Tačiau akivaizdu, kad jas galime prispausti prie bet kurio kampo. 

function parketas (ilg, kpl, kilg: integer): integer;

  var lent, post1, post2, eil, st: integer;

begin

  lent := 0; { paklotų lentelių skaičius }

  post1 := 0; { pirmosios horizontalios lentelės postūmis }

  { klojame horizontalias lenteles }

  for eil := 1 to kpl do

    begin

      lent := lent  + (kilg - post1 + ilg) div (2 * ilg);

      post1 := (post1 + 1) mod (2 * ilg);

    end;

  { klojame vertikalias lenteles }

  post2 := 0;

  for st := 1 to kilg do

    begin

      lent := lent + (kpl - post2 + ilg) div (2 * ilg);

      post2 := (post2 + 1) mod (2 * ilg);

    end;

  { ar sumažės lentelių ir pirmu ir antru atveju }

  if (((kilg + ilg) div (2 * ilg)) <> { lentelių skaičius pirmoje }

      ((kilg - post1 + ilg) div (2 * ilg))) { ir paskutinėje eilutėje }

     { prispaudus vertikalią lentelę prie kairiojo viršutinio kampo }

     and

     (((kpl + ilg) div (2 * ilg)) <> { lentelių skaičius pirmame }

      ((kpl - post2 + ilg) div (2 * ilg))) { ir paskutiniame stulpelyje }

     { prispaudus horizontalią lentelę prie kairiojo viršutinio kampo }

     then lent := lent - 1;

  parketas := lent;

end; { parketas }



Nesugalvojus įrodymo, galima perrinkti visus 2(ilg plokštelių dėliojimo variantus. 

Testo nr.�Pradiniai duomenys�Rezultatas�Paaiškinimai��1�10 10 10�1�Paprasčiausias atvejis��2�4 8 24�36�Kambario plotis lygus dvigubam lentelės ilgiui��3�5 15 20�48�Kambario plotis lygus trigubam lentelės ilgiui��4�5 16 20�52�Kambario plotis nesidalija iš lentelės ilgio��5�7 28 79�273�Vertikalioji plokštelė turi būti prispausta prie kampo��6�8 13 64�72�Horizontalioji plokštelė turi būti prispausta prie kampo��7�6 73 79�901�Bet kuri plokštelė gali būti prispausta prie kampo (plokštelių skaičius nesumažėja nei vienu atveju)��8�6 71 79�874�Bet kuri plokštelė gali būti prispausta prie kampo (plokštelių skaičius sumažėja abiem atvejais)��

60. Teiginys apie statųjį trikampį. Pradinius duomenis sugrupuosime pagal nežinomųjų (nulių) skaičių: 1) žinomi visų trijų kraštinių ilgiai; 2) žinomi dviejų kraštinių ilgiai; 3) žinomas tik vienos kraštinės ilgis; 3) nežinomas nė vienos kraštinės ilgis;

procedure sukeisti (var x, y: integer);

{ sukeičia vietomis dviejų kintamųjų reikšmes }

  var tarp: integer;

begin

  tarp := x;

  x := y;

  y := tarp

end; { sukeisti }



procedure trikampis (var a, b, c: integer; { trikampio kraštinių ilgiai }

                     var teig_a, teig_b, teig_c: boolean 

                     { ar teisingi atitinkami teiginiai });

  var než: integer; { nežinomų kraštinių skaičius }

begin

  { kiek kraštinių nežinome }

  než := ord(a=0) + ord(b=0) + ord(c=0);

  { bus patogiau, jei kraštinių ilgius surikiuosime nemažėjimo tvarka }

  if a > b then sukeisti(a, b);

  if a > c then sukeisti(a, c);

  if b > c then sukeisti(b, c);

  if než = 0 { jei žinome visus skaičius }

    then begin

           teig_a := a * a + b * b = c * c;

           teig_c := not teig_a

         end

    else begin

           teig_a := false;

           teig_c := false;

           case než of { tiek skaičių yra nežinomi }

             1: iešk_a (a, b, c, teig_b);

             2: iešk_ab (a, b, c, teig_b);

             3: iešk_abc (a, b, c, teig_b)

           end

         end

end; { trikampis }



Nors pirmasis atvejis jau išspręstas, jį galime trumpai panagrinėti. Jei iš trijų skaičių galima išrinkti tokius du, kad jų kvadratų suma lygi trečiojo kvadratų sumai, tuomet pradiniai duomenys tikrai išreiškia stačiojo trikampio kraštinių ilgius. Priešingu atveju teiginys klaidingas. Beje, sprendžiant ne tik šį atvejį, bet ir visą uždavinį, bus patogiau, jei prieš pradėdami nagrinėti atskirus atvejus, kraštinių ilgius surikiuosime nemažėjimo tvarka.

Sprendžiant likusius tris atvejus ne kartą teks traukti šaknį iš sveikojo skaičiaus. Paskalio kalboje yra tik šaknies iš realiųjų skaičių traukimo operacija. Kadangi veiksmai su realiaisiais skaičiais atliekami apytiksliai, nesame tikri, kad gautas rezultatas bus teisingas. Todėl reikia parašyti šaknies iš sveikojo skaičiaus traukimo funkciją. Skaičių, iš kurio trauksime šaknį, pažymėkime x. Parinkime du natūraliuosius skaičius a ir b, kurių intervale turėtų būti ieškoma šaknis, t. y. kad būtų teisinga nelygybė:

a ( (x < b

Pavyzdžiui, a = 1, b = x + 1. 

Šį intervalą padalykime pusiau ir imkime jo vidurį

ab := (a + b) div 2.

Ieškoma šaknis bus viename iš intervalų [a..ab[ ir [ab..b[. Kaip nustatyti kuriame intervale yra ieškoma šaknis? Skaičių x padalykime iš intervalo vidurio taško ab. Jei gautasis skaičius mažesnis už ab, tai šaknis bus  intervale [a..ab[, priešingu atveju – intervale [ab..b[. Tą intervalą kuriame yra šaknis, vėl dalykime pusiau, kol abu intervalo galai pasidarys gretimi (paeiliui einantys) skaičiai. Tuomet ieškoma šaknis bus apatinis intervalo rėžis (nes intervalo rėžius parinkome taip, kad a ( y < b.

function šaknis(x: integer): integer;

{ natūraliųjų skaičių kvadratinė šaknis }

  var a, b, { intervalas }

      ab,   { vienas iš naujo intervalo rėžių }

      xx: integer;

begin

  a := 1;  b := x + 1; { pradinis intervalas }

  while b - a > 1 do

    begin

      ab := (a + b) div 2;  { intervalas dalijamas pusiau }

      xx := x div ab;

      if xx >= ab then a := ab

                  else b := ab

    end;

  šaknis := a

end; { šaknis }



Kiek sunkiau patikrinti, ar teiginys gali būti teisingas, kai žinomi tik dviejų kraštinių ilgiai. Pažymėkime juos i1 ir i2, (i1 ( i2), o nežinomą ilgį – x. Galimi du variantai: i12 + i22 = x2 arba i22 - i12 = x2. Lieka patikrinti, ar žinomų kraštinių kvadratų suma (skirtumas) yra natūraliojo skaičiaus kvadratas.

procedure iešk_a (var a: integer; b, c: integer;

                  var teig_b: boolean);

{ ieško sprendinio, kai nežinomas vienas skaičius; b <= c }

  var t1, t2, a1: integer;

begin

  { galbūt trūkstamas skaičius yra įžambinė }

  t1 := b * b + c * c;

  a1 := šaknis (t1);

  if a1 * a1 = t1 { jei radome sprendinį }

     then teig_b := true

     else begin

            t2 := c * c - b * b;

            a1 := šaknis (t2);

            teig_b := a1 * a1 = t2

          end;

  if teig_b then a := a1

end; { iešk_a }



Imkime trečiąjį atvejį: žinomas tik vienos kraštinės ilgis. Perrinksime visus galimus antrosios kraštinės ilgio variantus: pradžioje ji bus lygi 1, po to 2 ir taip ją vis didinsime vienetu, kol rasime sprendinį arba perrinksime visus galimus atvejus. O kaip ieškoti trečiosios kraštinės žinant kitas dvi, jau mokame – ką tik nagrinėjome.

Kyla klausimas, kiek ilgai reikėtų ieškoti sprendinių? Juk reikia rasti mažiausio ploto trikampio kraštines. Galimi du variantai: žinomoji kraštinė yra arba statinys arba įžambinė. Jei randame sprendinį, ir sprendinys yra toks, kad žinoma kraštinė yra įžambinė, tai darbą baigiame. Mat šio trikampio plotas tikrai bus mažesnis už to, kuriame ši kraštinė būtų statinys. Žinomoji kraštinė gali būti įžambine tik tada, kai kitos dvi kraštinės yra už ją trumpesnės. Taigi variantų skaičius nėra didelis.

O jei toks sprendinys (t. y. kai žinomas kraštinė yra įžambinė) neegzistuoja tada pirmas rastasis sprendinys ir bus uždavinio rezultatas.

Jei duoto uždavinio sprendinio iš viso nėra (pvz., pradiniai duomenys 1, 0, 0), tai kyla klausimas kada reikėtų baigti darbą? Trikampio statinius pažymėkime a ir b, įžambinę – c. Klausimas, kada baigti paiešką gali iškilti tik tada, kai tikrai žinome, kad duota kraštinė gali būti tik trikampio statinis (ne įžambinė). tarkime, kad žinomoji kraštinė yra b. Trikampio įžambinės ieškome perrinkinėdami: c ( b, b + 1, b + 2, ... . Dar galime tikėtis surasti sprendinį, jei c2 - b2 ( (c- 1)2. Jei c2 - b2 > (c - 1)2, tai reiškia, kad toliau didinant įžambinę, nežinomas statinis bus iš intervalo ]c - 1, c[, t. y. nebebus sveikasis skaičius.

Taigi sprendinio reikia ieškoti, kol teisinga sąlyga c2 - b2 ( (c- 1)2.

procedure iešk_ab (var a, b: integer; c: integer;

                   var teig_b: boolean);

{ ieško sprendinio, kai nežinomi du skaičiai }

  var aa, bb, { gautas rezultatas, bet nežinia, ar optimalus, }

      a1, b1: integer;

begin

  aa := 0; bb := 0;      { dar neradome nė vieno sprendinio }

  b1 := 0;

  while  b1 < c do { gal turima kraštinė yra trikampio įstrižainė }

    begin

      b1 := b1 + 1;

      iešk_a (a1, b1, c, teig_b);

      if teig_b      { jei radome sprendinį }

      then if a1 < c    { ir jis optimalus }

           then begin

                  a := a1; b := b1;

                  exit

                end

           else if aa = 0 { neaišku, ar optimalus, }

                   then begin { tad reikia įsiminti }

                          aa := a1; bb := b1

                        end

    end; { while }

  { jei iki šios vietos radome sprendinį, tai jis tikrai optimalus }

  if teig_b

  then  begin

          a := aa; b := bb

        end

  else begin  { ieškome toliau, galbūt žinomas skaičius yra trikampio statinis }

         while not teig_b and 

                 (sqr(b1)-sqr(c) <= sqr(b1-1)) do

           begin

             b1 := b1 + 1;

             iešk_a (a1, c, b1, teig_b)

           end;

         if teig_b { jei radome sprendinį}

            then begin

                   a := a1; b := b1

                 end;

       end { else }

end; { iešk_ab }

Pagaliau nagrinėkime paskutinį atvejį: kai nežinoma nė viena kraštinė. Vėl perrenkame visus vienos kraštinės ilgio variantus: pradedame nuo 1. Spręsti uždavinį, kai žinomas vienos kraštinės ilgis, jau mokame. Akivaizdu, kad šiuo atveju sprendinys būtinai egzistuoja. 

procedure iešk_abc (var a, b, c: integer; 

                    var teig_b: boolean);

{ ieško sprendinio, kai nežinomos visos kraštinės }

begin

  c := 0;

  teig_b := false;

  while not teig_b do   { ciklas baigsis, nes sprendinys }

    begin                        { tikrai egzistuoja }

      c := c + 1;

      iešk_ab (a, b, c, teig_b)

    end

end; { iešk_abc }



Testo nr.�Pradiniai duomenys�Rezultatai�Paaiškinimai��1�15  9 12�TEIGINYS TEISINGAS�Žinomi visi trys skaičiai��2� 8  6  6�TEIGINYS KLAIDINGAS�Žinomi visi trys skaičiai��3� 9  0 12�TEIGINYS GALI BŪTI TEISINGAS

15 9 12�Žinomi du skaičiai, ieškomasis – įžambinė��4� 5  0  3�TEIGINYS GALI BŪTI TEISINGAS

4 3 5�Žinomi du skaičiai, ieškomasis – statinis��5� 3  6  0�TEIGINYS NEGALI BŪTI TEISINGAS�Žinomi du skaičiai, sprendinio nėra��6�10  0  0�TEIGINYS GALI BŪTI TEISINGAS

8 6 10�Žinomas vienas skaičius (įžambinė)��7� 0 28  0�TEIGINYS GALI BŪTI TEISINGAS

21 35 28�Žinomas vienas skaičius (statinis)��8� 7  0  0�TEIGINYS GALI BŪTI TEISINGAS

25 24 7�Žinomas vienas skaičius (statinis)��9� 1  0  0�TEIGINYS NEGALI BŪTI TEISINGAS�Žinomas vienas skaičius, sprendinio nėra��10� 0  0  0�TEIGINYS GALI BŪTI TEISINGAS

5 4 3�Nežinomas nė vienas skaičius��

61. Atstatyk skaičius. Keturis skaičius pažymėkime a, b, s ir mbk. Išnagrinėkime visus galimus variantus. Jų yra penki.

1. Žinomi visi keturi skaičiai. Tokiu atveju belieka patikrinti, ar duotieji skaičiai tenkina reikiamas sąlygas. Kadangi galioja lygybė mbk ( a div dbd ( b, tai nesunkiai galime rasti dviejų skaičių mažiausią bendrąjį kartotinį. Didžiausią bendrąjį daliklį galime surasti pagal Euklido algoritmą. Beje, sąlygoje nepasakyta, ką spausdinti, jei duoti keturi skaičiai ir jie nėra sprendinys. Susitarkime, kad tokiu atveju bus spausdinami keturi nuliai. 

2. Žinomi trys skaičiai. Turime išnagrinėti keturis variantus:

nežinomas a; jį randame taip: a ( s - b;

nežinomas b; sprendžiamas analogiškai, kaip pirmu atveju;

nežinomas s; jį randame taip: s ( a + b;

nežinomas mbk; turėdami skaičius a ir b randame jų mažiausią bendrąjį kartotinį.

Atstačius skaičių būtina patikrinti, ar tie keturi skaičiai yra sprendinys, mat trūkstamas skaičius atstatomas pagal kuriuos nors du skaičius, o pradiniai duomenys gali būti tokie, kad sprendinys neegzistuos. 

3. Žinomi du skaičiai. Turėsime nagrinėti net šešis variantus:

nežinomi skaičiai a ir b. Šiuo atveju skaičių a ir b teks ieškoti tarp mbk daliklių, taigi teks perrinkti visus daliklius. Sprendinys bus daliklis d1 ir skaičius (s - d1), jei šių skaičių mažiausias bendrasis kartotinis lygus mbk. 

nežinomi skaičiai s ir mbk. s ( a + b, mbk randamas įprastai;

nežinomi skaičiai a ir s. Žinome, kad galioja lygybė: �mbk div b ( a div dbd (a, b). Taigi skaičiai parenkami, kad galiotų lygybės: �a1 ( mbk div b, a ( a1 * dbd (a1, b);

nežinomi skaičiai b ir s. Sprendžiama analogiškai trečiam atvejui;

nežinomi skaičiai a ir mbk. Skaičiai parenkami taip: a ( b - s, �mbk apskaičiuojamas įprastai;

nežinomi skaičiai b ir mbk. Sprendžiama kaip ir penktuoju atveju.

Visais žemiau išvardytais atvejais (išskyrus, kai žinomi a ir b), atstačius trūkstamus skaičius, būtina patikrinti, ar turimi keturi skaičiai yra sprendinys. Mat du žinomi skaičiai gali būti tokie, kad sprendinys neegzistuoja.

4. Žinomas tik vienas skaičius. Teks išnagrinėti keturis atvejus: 

žinomas a: skaičių b parenkame lygų vienetui, o turėdami du skaičius, randame likusius du; 

žinomas b: skaičių a parenkame lygų vienetui, o turėdami du skaičius, randame likusius du; 

žinomas s: skaičius parenkame taip: a ( 1, b ( s - a, mbk įprastai;

žinomas mbk: skaičius parenkame taip: a ( 1, b ( mbk, s ( a + b.

5. Nežinomas nė vienas skaičius. Akivaizdu, kad šiuo atveju mažiausias galimas sprendinys yra toks: a ( 1, b ( 1, s ( 2, mbk ( 1; 

Beje, prieš atstatant skaičius būtina patikrinti, ar s ( a ir s ( b, jei tik šie skaičiai yra žinomi. Jei to nepatikrinę bandysime atstatyti, galime gauti neigiamą dėmenį. Taigi kad nereikėtų tos pačios sąlygos tikrinti daugelį kartų, patogiau tai padaryti vieną kartą prieš pradedant atstatinėti skaičius. 

function dbdf(x, y: integer): integer;

{ randa skaičių x ir y didžiausią bendrąjį daliklį }

begin

  if x = 0

     then dbdf := y

     else dbdf := dbdf(y mod x, x)

end; { dbdf }



function mbkf(x, y: integer): integer;

{ randa skaičių x ir y mažiausią bendrąjį kartotinį }

begin

  mbkf := x div dbdf(x, y) * y

end; { mbkf }



procedure netrūksta (var a, b, s, mbk: integer);

{ jei žinomi visi skaičiai, patikrina, ar jie teisingi }

begin

  if (a + b <> s) or (mbkf(a, b) <> mbk)

     then begin  { skaičių atstatyti nepavyko }

            a := 0; b := 0;

            s := 0; mbk := 0

          end

end; { netrūksta }



procedure vienas (var a, b, s, mbk: integer);

{ atstato skaičius, kai nežinomas vienas skaičius }

begin

  { atstatysime trūkstamą skaičių }

  if a = 0

  then a := s - b

  else if b = 0

       then b := s - a

       else if s = 0

            then s := a + b

            else mbk := mbkf(a, b);

  { patikrinsime, ar pavyko atstatyti }

  netrūksta (a, b, s, mbk)

end; { vienas }



procedure du (var a, b, s, mbk: integer);

{ atstato skaičius, kai nežinomi du skaičiai }

  var d, tarp: integer;

      rasta: boolean;

begin

  if (a = 0) and (b = 0)

  then begin { perrinksime mbk daliklius }

         d := 0;

         rasta := false;

         while (d <= mbk div 2) and not rasta do

           begin

             d := d + 1;

             if (mbk mod d = 0) and (s > d)

                then rasta := mbkf(s - d, d) = mbk

           end;

         if rasta

         then begin

                a := d;

                b := s - d

              end

       end

  else if (s = 0) and (mbk = 0)

       then begin

              s := a + b;

              mbk := mbkf(a, b)

            end

       else if mbk = 0

            then begin

                   if a = 0

                     then a := s - b

                     else b := s - a;

                   mbk := mbkf(a, b)

                  end

            else begin { nežinomas a arba b ir nežinomas s }

                   tarp := mbk div (a + b);

                   if a = 0

                     then a := tarp * dbdf(tarp, b)

                     else b := tarp * dbdf(a, tarp);

                   s := a + b

          end;

  { patikrinsime, ar pavyko atstatyti }

  netrūksta (a, b, s, mbk)

end; { du }



procedure trys (var a, b, s, mbk: integer);

{ atstato skaičius, kai nežinomi trys skaičiai }

begin

  { atstatysime skaičių a }

  if a = 0

     then a := 1;

  { atstatysime skaičių b }

  if s <> 0

     then b := s - a

     else if mbk <> 0

             then b := mbk

             else b := 1;

  { ir likusius du skaičius }

  if s = 0

     then s := a + b;

  if mbk = 0

     then mbk := mbkf(a, b);

  netrūksta (a, b, s, mbk)

end; { trys }



procedure keturi (var a, b, s, mbk: integer);

{ atstato skaičius, kai nežinomi visi keturi skaičiai }

begin

  a := 1; b := 1;

  s := 2; mbk := 1;

end; { keturi }



procedure atstatyti (var a, b, s, mbk: integer);

{ atstato trūkstamus skaičius }

  var trūksta,     { kiek skaičių reikia atstatyti }

      a1, b1, s1, mbk1: integer;

begin

  { rasime kiek skaičių trūksta }

  trūksta := ord(a=0) + ord(b=0) + ord(s=0) + ord(mbk=0);

  { išsaugosime pradinius duomenis }

  a1 := a; b1 := b; s1 := s; mbk1 := mbk;

  if (s <> 0) and ((a >= s) or (b >= s))

     then begin { dėmuo negali būti mažesnis už sumą }

            a := 0; b := 0;

            s := 0; mbk := 0

          end

     else case trūksta of

             0: netrūksta (a, b, s, mbk);

             1: vienas (a, b, s, mbk);

             2: du (a, b, s, mbk);

             3: trys (a, b, s, mbk);

             4: keturi (a, b, s, mbk)

           end;

  if (a = 0) and (trūksta <> 0)

     { jei nepavyko atstatyti, grąžiname ankstesnes reikšmes }

     then begin

            a := a1; b := b1;

            s := s1; mbk := mbk1

          end

end; { atstatyti }

II etapas

62. Klaida renkant skaičius. Tarkime, kad turėjome surinkti skaičių y, bet padarėme klaidą ir todėl gavome skaičių x. Ar tikrai klaida buvo padaryta pradanginant vieną skaičiaus skaitmenį? Ar galima jį atstatyti?

Sakykime, kad x ( 109, y ( 1069. Palyginame abiejų skaičių paskutinius skaitmenis ir kol jie lygūs, atmetinėjame juos. Mūsų pavyzdyje nuo galo atmetę sutampančius skaitmenis, gausime x ( 10, y ( 106. 

Dabar reikėtų iš teisingo skaičiaus (y) pabaigos atmesti vieną skaitmenį ir patikrinti, ar dabar abu skaičiai lygūs. Jei skaičiai lygūs, vadinasi, tikrai buvo praleistas vienas skaitmuo. Gali būti dar vienas atvejis, kai šitaip atsitinka. Jei atmetę vienodus skaitmenis gauname, kad x ( 0, tai gali būti, kad buvo pradangintas pirmasis skaitmuo. Tada reikėtų nuo y pabaigos atmesti visus nulius ir patikrinti, ar gautas skaičius yra vienaženklis.

O kaipgi gauti „teisingą“ ir „klaidingą“ skaičius? Reikia išnagrinėti keturis variantus.

Galbūt klaidos iš viso nebuvo padaryta, t. y. a + b ( c. Tokiu atveju daugiau nieko tikrinti nereikia.

Klaida padaryta surenkant pirmąjį skaičių. Tuomet ateisingas ( c - b.

Klaida padaryta surenkant antrąjį skaičių. Tuomet bteisingas ( c - a.

Klaida padaryta surenkant trečiąjį skaičių. Tuomet cteisingas ( a + b.

Šį kartą pateikiame ne tik procedūras, bet ir visą programą – vaizdžiau matysime, kaip gaunami atstatytieji skaičiai.

program klaida_skaičiuje;

type sktm = 0..10; { atstatytas skaitmuo }



  procedure skaitmuo (x, y: integer; var pamestas: sktm);

  { patikrina, ar renkant skaičių y nebuvo pamestas skaitmuo ir gautas skaičius x }

    var nuliai: integer;

  begin

    { skaitmenis pradėsime lyginti nuo galo }

    while x mod 10 = y mod 10

      begin

        x := x div 10;

        y := y div 10

      end;

    { numetame nulius skaičiaus y gale }

    nuliai := 0;

    while (y > 0) and (y mod 10 = 0) do

      begin

        y := y div 10;

        nuliai := nuliai + 1

      end;

    { surandame prarastą skaitmenį }

    pamestas := y mod 10;

    y := y div 10;

    { jei nepavyko sėkmingai atstatyti }

    if (x <> y) or (y > 9) and (nuliai <> 0)

        then pamestas := 10

  end; { skaitmuo }



  procedure atstatyti (a, b, c: integer; { pradiniai duomenys }

                     
 var as, bs, cs: sktm { atstatytieji skaitmenys });

  { atstato prarastus skaitmenis }

  begin

    { sėkmingai skaitmenų dar nepavyko atstatyti }

    as := 10; bs := 10; cs := 10;

    skaitmuo (c, a + b, cs);

    if c - b > 0

      then skaitmuo (a, c - b, as);

    if c - a > 0

      then skaitmuo (b, c - a, bs)

  end; { atstatyti }



  var f: text;

      a, b, c, i: integer;

      as, bs, cs: sktm;

begin

  assign(f, 'KLAIDA.DAT'); reset(f);

  for i := 1 to rink do

    begin

      readln(f, a, b, c);

      writeln(a: 6, b: 6, c: 6);

      if a + b = c { jei pradiniai duomenys teisingi }

      then writeln(a: 6, b: 6, c: 6)

      else begin

             atstatyti (a, b, c, as, bs, cs);

             if (cs >= 10) and (as >= 10) and (bs >= 10)

                 then writeln('SPRENDINIŲ NĖRA');

             if cs < 10

                then writeln(a: 6, b: 6, a+b: 6);

             if as < 10

                then writeln(c-b: 6, b: 6, c: 6);

             if bs < 10

                then writeln(a: 6, c-a: 6, c: 6)

           end;

      writeln

    end

end.

Sąlygoje buvo prašoma prarašyti programą, kuri spręstų uždavinį su šešiais duomenų rinkiniais. Testuose pateikti tik duomenų rinkiniai, nenurodant, į kurį testą jie įeina.



63. Paprastosios trupmenos. Sugeneruoti visas paprastąsias trupmenas nesudėtinga – tereikia išbandyti visus galimus skaitiklius ir vardiklius nuo 1 iki n. Nors sąlygoje nėra aiškiai pasakyta, bet iš pavyzdžio matyti, kad reikia spausdinti tik skirtingas trupmenas, t. y. jei radome trupmeną 1/2, tai 2/4 spausdinti nebereikia.

Nesunku pastebėti, kad reikia spausdinti tik tas trupmenas, kurių negalima suprastinti. Jei suprastinę trupmeną gausime nebesuprastinamą trupmeną, jos vardiklis dar paklius į intervalą [2..n], vadinasi, ji ir taip bus išspausdinta.

function dbd(x, y: integer): integer;

{ skaičių x ir y didžiausias bendrasis daliklis }

begin

  while (x <> 0) and (y <> 0) do

    if x >= y

       then x := x mod y

       else y := y mod x;

    dbd := x + y

end;  { dbd }



procedure spausdinti (n :integer);

  var sk, vard: integer;

begin

  for vard := 2 to n do

    for sk := 1 to vard-1 do

      if dbd(sk, vard) = 1

        then write (sk, '/', vard, ' ');

end; { spausdinti }

III etapas

64. Dalikliai ir kartotiniai. Tai nesunkus uždavinys, tereikia išnagrinėti visus galimus variantus. Jų yra penki.

1. Žinomi visi keturi skaičiai. Tokiu atveju pagal duotas funkcijas belieka patikrinti, ar duotieji skaičiai tenkina reikiamas sąlygas;

2. Žinomi trys skaičiai:

Nežinomas a. Pagal formulę a ( mbk div b ( dbd randame trūkstamą skaičių. Tačiau pagal šią formulę galima rasti skaičius tik tada, jei b, dbd ir mbk tikrai yra korektiški pradiniai duomenys – t. y. jei egzistuoja sprendinys. Todėl suradus skaičių a, būtina patikrinti, ar turimi keturi skaičiai yra sprendinys.

Nežinomas b. Sprendžiamas analogiškai tam atvejui, kai nežinomas a.

Nežinomas dbd. Turėdami skaičius a ir b randame jų didžiausią bendrąjį daliklį ir patikriname ar mbk tikrai yra skaičių a ir b mažiausias bendrasis kartotinis.

Nežinomas mbk. Turėdami skaičius a ir b randame jų mažiausią bendrąjį kartotinį ir patikriname ar dbd tikrai yra skaičių a ir b didžiausias bendrasis daliklis.

3. Žinomi du skaičiai. Visais žemiau išvardytais atvejais (išskyrus, kai žinomi a ir b), atstačius trūkstamus skaičius, būtina patikrinti ar turimi keturi skaičiai yra sprendinys. Mat du žinomi skaičiai gali būti tokie, kad sprendinys neegzistuoja. 

Nežinomi skaičiai a ir b . Skaičių a prilyginame dbd, o b gauname pagal formulę b ( mbk div a ( dbd. Skaičių a galima būtų prilyginti ir vienetui, tačiau tokiu atveju gerokai padidėja perpildymo tikimybė.

Nežinomi skaičiai dbd ir mbk. Jie randami pasinaudojus sąlygoje pateiktais algoritmais.

Nežinomi skaičiai a ir dbd. Parenkame: a ( mbk, ir pasinaudodami duotąja sąlygoje funkcija randame dbd.

Nežinomi skaičiai b ir dbd. Sprendžiama analogiškai aukščiau aprašytam atvejui.

Nežinomi skaičiai a ir mbk. Skaičius parenkame taip: a ( dbd, mbk apskaičiuojame pasinaudodami sąlygoje pateikta funkcija mbkf.

Nežinomi skaičiai b ir mbk. Sprendžiama analogiškai aukščiau aprašytam atvejui.

4. Žinomas tik vienas skaičius. Tokiu atveju visi nežinomi skaičiai prilyginami žinomam šitokiu būdu: 

Žinomas a. Skaičių b parenkame bet kokį (pavyzdžiui, lygų a) ir pagal juos apskaičiuojame dbd bei mbk. Tokiu atveju jie sutaps su a. 

Žinomas b. Skaičių a parenkame bet kokį (pavyzdžiui, lygų b) ir pagal juos apskaičiuojame dbd bei mbk. Tokiu atveju jie sutaps su b. 

Žinomas dbd. Skaičius a ir b parenkame vienodus: a ( b ( dbd, o tuomet apskaičiuojame mbk. Jis, beje, taip pat lygus mbk.

Žinomas mbk.Skaičius a ir b parenkame vienodus: a ( b ( mbk, o tuomet apskaičiuojame dbd. Jis, beje, taip pat lygus dbd.

Nežinomas nė vienas skaičius. Akivaizdu, kad šiuo atveju tinka bet kuris galimas sprendinys. Pasirinkime tokį: a ( 1, b ( 1, dbd ( 1, mbk ( 1.

Pateikiame išbaigtą algoritmą.

procedure netrūksta (var a, b, dbd, mbk: integer);

{ jei žinomi visi skaičiai, patikrina, ar jie teisingi }

begin

  if (dbdf(a, b) <> dbd) or (mbkf(a, b) <> mbk)

     then begin  { skaičių atstatyti nepavyko }

            a := 0; b := 0;

            dbd := 0; mbk := 0

          end

end; { netrūksta }



procedure vienas (var a, b, dbd, mbk: integer);

{ atstato skaičius, kai nežinomas vienas skaičius }

begin

  { atstatysime trūkstamą skaičių }

  if a = 0

  then a := mbk div b * dbd

  else if b = 0

          then b := mbk div a * dbd

          else if dbd = 0

                  then dbd := dbdf(a, b)

                  else mbk := mbkf(a, b);

  { patikrinsime, ar pavyko atstatyti }

  netrūksta (a, b, dbd, mbk)

end; { vienas }



procedure du (var a, b, dbd, mbk: integer);

{ atstato skaičius, kai nežinomi du skaičiai }

begin

  if (a = 0) and (b = 0)

  then begin a := dbd;

             b := mbk div a * dbd

       end

  else if (dbd = 0) and (mbk = 0)

       then begin dbd := dbdf(a, b);

                  mbk := mbkf(a, b)

            end

       else if mbk = 0

            then begin if a = 0

                          then a := dbd

                          else b := dbd;

                       mbk := mbkf(a, b)

                 end

            else begin if a = 0

                          then a := mbk

                          else b := mbk;

                       dbd := dbdf(a, b)

                 end;

  { patikrinsime, ar pavyko atstatyti }

  netrūksta (a, b, dbd, mbk)

end; { du }



procedure trys (var a, b, dbd, mbk: integer);

{ atstato skaičius, kai nežinomi trys skaičiai }

  var žin: integer;

begin

  { randame žinomą skaičių }

  žin := a + b + dbd + mbk;

  a := žin; b := žin;

  dbd := žin; mbk := žin

end; { trys }



procedure keturi (var a, b, dbd, mbk: integer);

{ atstato skaičius, kai nežinomi visi keturi skaičiai }

begin

  a := 1; b := 1;

  dbd := 1; mbk := 1

end; { keturi }



procedure atstatyti (var a, b, dbd, mbk: integer);

{ atstato trūkstamus skaičius }

  var trūksta: integer; { kiek skaičių reikia atstatyti }

begin

  { rasime kiek skaičių trūksta }

  trūksta := ord(a=0) + ord(b=0) + ord(dbd=0) + ord(mbk=0);

  case trūksta of

    0: netrūksta (a, b, dbd, mbk);

    1: vienas (a, b, dbd, mbk);

    2: du (a, b, dbd, mbk);

    3: trys (a, b, dbd, mbk);

    4: keturi (a, b, dbd, mbk)

  end

end; { atstatyti }

65. Lygtis. Sprendimas jau yra pasufleruotas sąlygoje. Ten pasakyta, kad galioja lygybė:

anxn + an-1xn-1 + ... + a1x + a0 = an(x - x1)(x - x2)...(x - xn-1)(x - xn);

čia x1, x2, ... xn – lygties šaknys.

Nesunku pastebėti, kad koeficientas prie nulinio laipsnio a0 yra šitoks:

a0 = an ( (x1 ( x2 ( ... ( xn-1 ( xn).

O tai reiškia, kad visos ieškomos lygties šaknys pakliūva į intervalą  [-|a0|,  |a0|]. Belieka perrinkti visus skaičius iš šio intervalo ir patikrinti, ar jie nėra lygties šaknys. Norint algoritmą padaryti dar efektyvesnį, galima būtų perrinkti tik a0 daliklius, priklausančius minėtam intervalui. Tačiau mūsų pateikiamame algoritme to nėra.

Gali susidaryti atskiras atvejis, kai a0 ( 0. Tokiu atveju lygtis visuomet turės vieną šaknį – nulį ir ją bus galima užrašyti tokiu pavidalu:

x(anxn - 1 + an-1xn-2 + ... + a1).

Iškėlę x prieš skliaustus gauname vienu laipsniu mažesnę lygtį, kurią spręsti jau mokame.

Sąlygoje nepasakyta, ar reikia spausdinti vienodus lygties sprendinius. Susitariame spausdinti tik skirtingus lygties sprendinius.

const MAXN = 50; { maksimalus lygties laipsnis }

type mas = array [0..MAXN] of integer;



procedure spręsti (n: integer;        { lygties laipsnis }

                   a: mas;            { jos koeficientai }

                   var spr: mas;      { lygties sprendiniai }

                   var kiek: integer  { jų skaičius });

{ randa sveikuosius lygties sprendinius }

  var i, j, rez, x: integer;

begin

  kiek := 0; { dar nerasta nė vieno sprendinio }

  { patikrinsime, ar nulis yra lygties sprendinys }

  i := 0;

  while a[i] = 0 do

    i := i + 1;

  if i <> 0

     then begin      { sumažinsime lygties laipsnį }

             kiek := kiek + 1;

             spr[kiek] := 0;

             for j := 0 to n - i do

               a[j] := a[j + i];

             n := n - i

          end;

  { ieškosime kitų lygties sprendinių }

  for x := -abs(a[0]) to abs(a[0]) do

    begin { tikrinsime, ar x yra lygties sprendinys }

      rez := a[n];

      for i := n-1 downto 0 do

        rez := rez * x + a[i];

      if rez = 0

         then begin

                kiek := kiek + 1;

                spr[kiek] := x

              end

    end

end; { spręsti }



Lygties nr.�Lygties koeficientai�Rezultatai�Paaiškinimai��1�1 1 -32000�32000�x - 32000 ( 0��2�2 1 1 -10100�100 -101�x2 + x - 10100 ( 

(x - 100) (x + 101)��3�4 1 -10 35 -50 24�1 2 3 4�x4 - 10x3 + 35x2 - 50x + 24 ( 

(x - 1)(x - 2)(x - 3)(x - 4)��4�5 1 0 -5 0 4 0�-2 -1 0 1 2�x5 - 5x3 + 4x ( 

x(x + 2)(x + 1) (x - 2)(x - 1)��5�10 3 -3 -3 -3 -3 -3 3 3 3 2 1�1���6�10 3 -3 -3 -3 -3 -3 3 3 3 3 0�0 1���7�37 1 1 ...1 

     36 vienetai �-1���8�37 1 1 ... 1 0

     36 vienetai �0���Sąlygoje prašoma parašyti programą, kuri iš karto išspręstų aštuonias lygtis. Aukščiau yra pateiktas vienas testas su aštuoniomis lygtimis.



66. Domino kauliukai. Iš duotų kauliukų sudarykime grafą. Skaičius, užrašytus ant kauliuko pusių, laikysime grafo viršūnėmis. Iš viso grafas turės 7 viršūnes, sunumeruotas nuo 0 iki 6. Kiekvieną kauliuką laikysime briauna. Briauna jungs tas dvi viršūnes, kurių numeriai užrašyti ant abiejų kauliuko šonų. Sąlygoje pateiktus duomenis galima pavaizduoti tokiu grafu (66.2 pav.).

66.2 pav.

Kauliukų išdėliojimas į vieną liniją atitinka kelią, kai visos grafo briaunos apeinamos po vieną kartą. 

Jei grafas yra jungus, ir jame egzistuoja toks kelias, kad visos briaunos jame būtinai sutinkamos ir tik vieną kartą, tas kelias vadinamas Oilerio keliu.

Oilerio teorema. Grafas turi Oilerio kelią tada ir tik tada, jei grafas yra jungus ir jame arba nėra nelyginio laipsnio viršūnių, arba yra lygiai dvi. Viršūnės laipsnis – tai iš grafo viršūnės išeinančių briaunų skaičius.

Nors ši teorema skamba labai moksliškai, nesunku ir patiems ją suformuluoti. Jei į kurią nors viršūnę atėjome, iš jos būtinai reikės išeiti. Jei briaunų skaičius nelyginis, tai kažkurį kartą atėję, nebegalėsime išeiti. Na, o dvi viršūnės gali būti nelyginių laipsnių. Mat į pirmąją viršūnę nereikia ateiti pradedant vaikščioti, o iš paskutiniosios nereikia išeiti.

Taigi norint patikrinti, ar galima sudėlioti domino kauliukus reikia suskaičiuoti, kiek kartų kiekvienas skaičius iš intervalo [0, 6] užrašytas ant domino kauliukų. Reikia nepamiršti patikrinti, ar grafas jungus, t. y. ar nesusidarys tokia situacija kaip 66.3 pav.

66.3 pav. 

Jungumą patikrinti nesunku. Pradėję eiti nuo bet kurios viršūnės, randame viršūnes, į kurias galime pakliūti tiesiogiai arba per tarpininkus. Jei iš pradinės viršūnės galime pakliūti į visas šešias viršūnes, grafas jungus.

type domino = array [0..6, 0..6] of integer;

     v_aibė = set of 0..6;      { viršūnių aibė }

     v_laipsniai = array [0..6] of integer;



procedure eiti (v: integer; { viršūnė, nuo kurios pradedame eiti }

                dm: domino;

                var aibė: v_aibė);

{ randa visas viršūnes, pasiekiamas iš pradinės viršūnės }

  var i: integer;

begin

  aibė := aibė + [v];

  for i := 0 to 6 do

    if (dm[i, v]  <> 0) and not (i in aibė)

       then eiti (i, dm, aibė)

end; { eiti }



function rasti (dm: domino): boolean;

{ nustato, ar galima išdėlioti kauliukus į vieną liniją }

  var s: v_laipsniai;  { akučių pasikartojimų skaičius, t. y. viršūnių laipsniai }

      aibė: v_aibė;    { viršūnių, pasiekiamų iš nulinės viršūnės, aibė }

      nel, i, j: integer;

begin

  { rasime grafo viršūnių laipsnius }

  for i := 0 to 6 do

    s[i] := 0;

  for  i := 0 to 6 do

    for j :=  i to  6 do

      begin

        s[i] := s[i] + dm[i, j];

        s[j] := s[j] + dm[i, j]

      end;

  { rasime nelyginių viršūnių skaičių }

  nel := 0;

  for i := 0 to 6 do

    nel := nel + s[i] mod 2;

  { tikrinsime, ar grafas jungus }

  aibė := [];

  { į aibę įtrauksime izoliuotas viršūnes }

  { jų apeiti nereikia }

  for i := 0 to 6 do

    if s[i] = 0

       then aibė := aibė + [i];

  eiti (0, dm, aibė); { rasime visas viršūnes, kurias galime pasiekti iš nulinės }

  if ((nel = 2) or (nel = 0)) and (aibe = [0..6])

     then rasti := true

     else rasti := false

end; { rasti }



Testo nr.�Rezultatas�Komentaras��1�GALIMA�Duoti 5 kauliukai; dvi viršūnės su nelyginiu laipsniu, trys izoliuotos��2�GALIMA�Duota 13 kauliukų; dvi viršūnės su nelyginiu laipsniu, izoliuotų viršūnių nėra��3�GALIMA�Duota 14 kauliukų; visų viršūnių laipsniai lyginiai, izoliuotų viršūnių nėra��4�NEGALIMA�Duota 12 kauliukų; keturios viršūnės turi nelyginius laipsnius, izoliuotų viršūnių nėra��5�NEGALIMA�Duota 12 kauliukų: dvi viršūnės turi nelyginius laipsnius, izoliuotų viršūnių nėra, grafas nejungus��6�NEGALIMA�Duoti 9029 kauliukai; dvi viršūnės turi nelyginius laipsnius, izoliuotų viršūnių nėra, grafas nejungus��7�NEGALIMA�Duoti 18994 kauliukai; keturios viršūnės turi nelyginius laipsnius, izoliuotų viršūnių nėra��8�GALIMA�Duoti 18995 kauliukai; dvi viršūnės turi nelyginius laipsnius, izoliuotų viršūnių nėra��

67. Daugianariai. Operacijos su daugianariais apibrėžtos sąlygoje. Paprasčiausios – sudėties ir atimties operacijos. Sudedant (atimant) daugianarius, sudedami (atimami) koeficientai prie atitinkamų laipsnių. Rezultato laipsnį prilyginame daugianario su aukštesniu laipsniu laipsniui. 

Sąlygoje prašoma skaičiuoti vieno skaitmens po kablelio tikslumu. Todėl reikėtų rezultato laipsnį mažinti tol, koeficientas prie daugianario laipsnio taps didesnis už nurodytą tikslumą, pavyzdžiui, jei atėmę du daugianarius gavome rezultatą:

0,001x3 + 10x2 - 0,05x + 16

tai rezultato laipsnis bus lygus 2, o koeficientas prie x – nuliui.

const  NMAX = 10;      { didžiausias daugianario laipsnis }

       EPSILON = 0.1;  { skaičiavimų tikslumas }

type daug = record     { daugianaris }

              n: 0..NMAX;                { daugianario laipsnis }

              k: array [0..NMAX] of real { ir koeficientai }

            end;



function max(x, y: integer): integer;

{ randa didesnįjį iš dviejų skaičių }

begin

  if x < y

    then max := y

    else max := x

end; { max }



procedure normal (var c: daug);

{ perskaičiuoja daugianario c laipsnį }

{ skaičiavimai atliekami epsilon tikslumu }

begin

  while (abs (c.k[c.n]) < epsilon) and (c.n > 0) do

    c.n := c.n - 1

end; { normal }



procedure sudėtis (a, b: daug; var c: daug);

{ sudeda daugianarius a ir b }

  var i: 0..NMAX;

begin

  for i := 0 to NMAX do

    begin

      c.k[i] := 0;

      if i <= a.n

         then c.k[i] := c.k[i] + a.k[i];

      if i <= b.n

         then c.k[i] := c.k[i] + b.k[i]

    end;

  { sumos laipsnis lygus didesniajam iš daugianarių laipsniui }

  c.n :=  max (a.n, b.n);

  { c laipsnis perskaičiuojamas vieno skaitmens po kablelio tikslumu }

  normal(c)

end; { sudėtis }



procedure atimtis (a, b: daug; var c: daug);

{ iš daugianario a atima b }

  var i: 0..NMAX;

begin

  for i := 0 to NMAX do

    begin

      c.k[i] := 0;

      if i <= a.n

         then c.k[i] := c.k[i] + a.k[i];

      if i <= b.n

         then c.k[i] := c.k[i] - b.k[i]

    end;

  { skirtumo laipsnis lygus didesniajam iš daugianarių laipsniui }

  c.n := max (a.n, b.n);

  { c laipsnis perskaičiuojamas vieno skaitmens po kablelio tikslumu }

  normal(c)

end; { atimtis }



Kiek sudėtingiau atliekama daugyba. Patogiausia dauginti po vieną narį. Pavyzdžiui, 

(2,5x3 + 1) ( (2x2 - 6x) = (5x5 - 15x4 + 2x2 - 6x)

Pradžioje rezultatą prilyginkime nuliui. Beje, rezultato laipsnis bus lygus abiejų daugianarių laipsnių sumai. 

x laipsnis�x5�x4�x3�x2�x1�x0��koeficientas prie šio laipsnio�0,0�0,0�0,0�0,0�0,0�0,0��

Imkime mažiausiąjį pirmojo daugianario narį (t. y. koeficientą prie x0) – 1, ir padauginkime jį iš antrojo daugianario (t. y. visų jo narių). Sudauginant koeficientai susidaugina, o laipsniai susideda. Gausime 2x2 - 6x. Šią sandaugą pridėkime prie skaičiuojamo rezultato:

x laipsnis�x5�x4�x3�x2�x1�x0��koeficientas prie šio laipsnio�0,0�0,0�0,0�2,0�-6,0�0,0��

Pirmojo daugianario koeficientai prie x1 ir x2 lygūs nuliui. Imkime narį 2.5x3. Jį dauginkime iš visų antrojo daugianario narių. Gausime 5x5 - 15x4. Gautą sandaugą pridedame prie rezultato:

x laipsnis�x5�x4�x3�x2�x1�x0��koeficientas prie šio laipsnio�5,0�-15,0�0,0�2,0�-6,0�0,0��

Gavome, kad sandauga lygi 5x5 - 15x4 + 2x2 - 6x.

Beje, nebūtina pirma padauginti vieną daugianario narį iš antrojo daugianario ir tik tuomet sandaugą pridėti prie rezultato. Sudauginus du daugianarių narius, jų sandaugą galima iš karto pridėti prie rezultato.

procedure daugyba (a, b: daug; var c: daug);

{ sudaugina daugianarius a ir b }

  var i, j: 0..NMAX;

begin

  for i := 0 to NMAX do

    c.k[i] := 0;

  { dauginant rezultato laipsnis lygus dauginamųjų laipsnių sumai }

  c.n := a.n + b.n;

  for i := 0 to a.n do

      for j := 0 to b.n do

        { dauginant laipsniai sudedami, koeficientai susidauginami }

        c.k[i+j] := c.k[i+j] + a.k[i] * b.k[j];

  { c laipsnis perskaičiuojamas vieno skaitmens po kablelio tikslumu }

  normal(c)

end; { daugyba }



Sudėtingiausias iš visų veiksmų yra dalyba. Žemiau (67.1 pav.) parodytas pavyzdys, kaip ji atliekama (Dalijama stulpeliu. Sakykime, kad reikia atlikti tokį veiksmą: (2,5x5 +3x + 1) / (2x2 - 6x).

67.1 pav.

procedure dalyba (a, b: daug; var c: daug);

  var t: daug;

      i: 0..NMAX;

begin

  for i := 0 to NMAX do

    c.k[i] := 0;

  { dalijant daugianarių laipsniai atsiima }

  c.n := a.n - b.n;

  while a.n >= b.n do

    begin

      t.n := a.n - b.n;

      for i := 0 to NMAX do

        t.k[i] := 0;

      t.k[t.n] := a.k[a.n]/b.k[b.n];

      sudėtis (c, t, c);

      daugyba (b, t, t);

      atimtis (a, t, a)

    end;

    normal(c)

  end; { dalyba }





  procedure dal_liek (a, b: daug; var c: daug);

    var dalmuo, sandauga: daug;

  begin

    dalyba (a, b, dalmuo);

    daugyba (b, dalmuo, sandauga);

    atimtis (a, sandauga, c);

    normal(c)

  end; { dal_liek }



�Opera-cijos Nr.�Pradiniai duomenys�Rezultatai�Paaiškinimai��1�+

8  4.0  -5.0  0.5  3.5  3.3  0.0  -1.0  7.6  10.0

7  9.0  3.5  0.5  0.7  4.0  5.0  -3.6  -10.0�8  4.0  4.0  4.0  4.0  4.0  4.0  4.0  4.0  0.0�(4x8 - 5x7 + 0,5x6 + 3,5x5 + 3,3x4- x2 + 7,6x + 10) + 

(9x7 + 3,5x6 + 0,5x5 + 0,7x4 + 4x3 + 5x2 - 3,6x - 10) = 

4x8 + 4x7 + 4x6 + 4x5 + 4x4+ 4x3 + 4x2 + 4x

��2�-

3  -2.0  -3.0  4.4  22.0

3  -2.0  -3.0  4.4  22.0�0   0.0�(-2x3 - 3x2 +4,4x + 22) - 

(-2x3 - 3x2 +4,4x + 22) = 0��3�*

7  7.0  -7.0  7.0  7.0  7.0  7.0  7.0  7.0

3  3.0  3.0  -3.0  3.0�10  21.0  0.0  -21.0  84.0  0.0   42.0  42.0  42.0  21.0  0.0  21.0�(7x7 - 7x6 + 7x5 + 7x4 + 7x3 + 7x2 + 7x + 7) * 

(3x3+ 3x2 - 3x + 3) = 

(21x10 - 21x8 +84x7 + 42x5 + 42x4 + 42x3 + 21x2 + 21

��4�/

10  21.0  0.0  -21.0  84.0  0.0  42.0  42.0  42.0  21.0  0.0  21.0

7  7.0  -7.0  7.0  7.0  7.0  7.0  7.0  7.0�3  3.0  3.0  -3.0  3.0�(21x10 - 21x8 + 84x7 + 42x5 + 42x4+ 42x3 + 21x2 + 21) / 

(7x7 - 7x6 + 7x5 + 7x4 + 7x3 + 7x2 + 7x + 7) = 

3x3 + 3x2 - 3x + 3

��5�\

10  21.0  0.0  -21.0  84.0  0.0  42.0  42.0  42.0  21.0  0.0  21.0

7  7.0  -7.0  7.0  7.0 7.0  7.0  7.0  7.0�0  0.0�(21x10 - 21x8 + 84x7 + 42x5 + 42x4+ 42x3 + 21x2 + 21) \ 

(7x7 - 7x6 + 7x5 + 7x4 + 7x3 + 7x2 + 7x + 7) = 0

��6�\

10  21.0  0.0  -21.0  84.0  10.0  42.0  42.0 42.0  21.0  0.0  21.0

7  7.0  -7.0  7.0  7.0  7.0  7.0  7.0  7.0�6  10.0  0.0  0.0  0.0  0.0  0.0  0.0�(21x10 - 21x8 + 84x7 + 10x6+ 42x5 + 42x4+ 42x3 + 21x2 + 21) \ 

(7x7 - 7x6 + 7x5 + 7x4 + 7x3 + 7x2 + 7x + 7) = 

10x6��7�/

4  5.0  1.0  4.0  1.0  1.0

2  3.0  7.0  3.0�2  1.7  -3.6  8.0�(5x4 + x3 + 4x2 + x + 1) / 

(3x2 +7x + 1) = 

1,7x2 -3,6x + 8��8�\

4  5.0  1.0  4.0  1.0  1.0

2  3.0  7.0  3.0�1 -44.1 -22.9�(5x4 + x3 + 4x2 + x + 1) \ 

(3x2 +7x + 1) = 

-44,1x - 22,9���Sąlygoje buvo prašoma parašyti programą, kuri atliktų aštuonis veiksmus. Aukščiau yra pateiktas vienas testas, su aštuoniomis operacijomis, kurias reikia atlikti. 



68. Hammingo seka. Vienas galimų šio uždavinio sprendimo būdų – rasti ribines i, j ir k reikšmes, sugeneruoti visus sekos narius ir surikiuoti juos didėjimo tvarka. 

Tačiau yra ir geresnis algoritmas. Sekos narius galime iš karto generuoti didėjimo tvarka. Naujus generuosime daugindami jau sugeneruotus narius iš tų trijų daugiklių. Pradžioje pateikiame tokią lentelę:

Hammingo seka             Narys, padaugintas iš atitinkamo daugiklio

1                                          2         3         5

2                                          4        6        10

3                                          6        9        15

4                                          8       12        20

5                                        10       15        25

Pradžioje imame pirmąjį Hammingo sekos narį (vienetą), jį įtraukiame į rezultatų seką. Tą narį padauginame iš trijų daugiklių. Gauname tris skaičius 2, 3, 5. Imame mažiausią iš jų – 2. Tai bus antrasis sekos narys. Tolesnį sekos narį turime rinktis iš 3, 5 ir 4, 6, 10. Tačiau iš dviejų paskutinių skaičių rinktis nėra prasmės: mat pirmasis Hammingo sekos narys, padaugintas iš kažkurio daugiklio, tikrai yra mažesnis už antrąjį sekos narį padaugintą iš to paties daugiklio, pavyzdžiui:

h1 ( 3 ( h2 ( 3, t. y. 

1 ( 3 ( 2 ( 3

Todėl pakanka rinktis iš 4, 3, 5. Mažiausias iš šių skaičių yra 3. Tai bus trečiasis Hammingo sekos narys. Taigi kiekvieną kartą renkamės iš trijų skaičių. Iš tų skaičių išrenkame mažiausią ir įtraukiame į Hammingo seką, jeigu jis dar nebuvo įtrauktas anksčiau. Įtrauktąjį skaičių pakeičiame tolesniu jau sugeneruotu Hammingo sekos nariu, padaugintu iš to paties daugiklio. 

Šiuos veiksmus kartojame tol, kol paspaudžiamas tarpo klavišas arba viršijamas maxlongint.

Beje, rašant algoritmą, patogiau yra saugoti ne tris Hammingo sekos narius, padaugintus iš daugiklio, o jų numerius. Tokiu atveju įtraukus kurį nors narį į seką, patogu pereiti prie vienu numerio didesnio nario.

const MAX = 2000; { didžiausias galimas Hammingo sekos ilgis }

      d: array [1..3] of integer = (2, 3, 5); { sekos daugikliai }

type masyvas = array [1..MAX] of longint;



procedure generuoti (var seka: masyvas; { Hammingo seka}

                     var kiek: integer  { jos narių skaičius });

  var a: array [1..3] of integer;    { trijų Hammingo sekos narių }

                           { numeriai, iš kurių rinksimės tolesnį narį }

      liko: integer; { iš tiek daugiklių dar galima dauginti tris tiriamus }

                        { sekos narius ir dar nebus viršytas maxlongint }

      i, min, num: longint;

begin

  { sugeneruojame pirmąjį narį }

  kiek := 1; seka[1] := 1;

  for i := 1 to 3 do

     a[i] := 1;

  liko := 3;

  while not keypressed and (liko <> 0) do

    { kol nepaspaustas tarpo klavišas ir neviršijamas maxlongint }

    begin

      { iš trijų narių, padaugintų iš daugiklių, išrinksime mažiausią }

      min := maxlongint;

      for i := 1 to 3 do

        if (a[i] <> 0) and (seka[a[i]] * d[i] < min)

          then begin

                 min := seka[a[i]] * d[i];

                 num := i

               end;

      { jei narys neįtrauktas į seką }

      if min > seka[kiek]

         then begin

                kiek := kiek + 1;

                seka[kiek] := min

              end;

      { pereinama prie tolesnio sekos nario }

      a[num] := a[num] + 1;

      { tikrinsime, ar tolesnis sekos narys neviršys maxlongint }

      if maxlongint div d[num] < seka[a[num]]

         then begin

                a[num] := 0;

                liko := liko - 1

              end

    end

end; { generuoti }



69. Trumpiausias kelias. Prieš pradėdami spręsti uždavinį, išnagrinėsime Dijkstros algoritmą. 

Tarkime, kad duoti galimų maršrutų punktai ir juos jungiantys vienos ar abiejų krypčių keliai. Kiekvienam keliui nurodyta tam tikra vertė (svoris). Tai neneigiamas skaičius, kuris paprasčiausiu atveju gali reikšti kelio ilgį, bet gali apibūdinti ir kelyje sugaištą laiką, benzino išlaidas ar kelis parametrus iš karto. Reikia rasti tokį maršrutą tarp dviejų punktų, kad juos jungiančių atkarpų (kelių) verčių suma būtų mažiausia. Dijkstros algoritmas sprendžia šį uždavinį.

Apibrėžkime uždavinį formaliai: grafą G sudaro viršūnių aibė V ir briaunų aibė B: G ( (V, B). Grafas yra svorinis, t. y. kiekvienai jo briaunai yra priskirta vertė (svoris) s(i, j) ( 0, čia i, j – viršūnių, kurias jungia atitinkama briauna, numeriai. Duota viršūnė Alfa ( V. Reikia rasti trumpiausią kelią nuo viršūnės Alfa iki visų likusių grafo viršūnių.

Iš pradžių nežinome koks yra trumpiausias atstumas nuo Alfa iki likusių viršūnių, nežinome net ar tos viršūnės yra pasiekiamos iš Alfa, todėl atstumus nuo Alfa iki visų likusių viršūnių prilyginame (. Trumpiausias kelias nuo Alfa iki jos pačios tikrai lygus nuliui (69.2.a pav.).

69.2.a pav.

Sudarome dar neišnagrinėtų viršūnių aibę. Ją pažymėkime Q. Pradžioje į ją sudedame visas viršūnes. Viršūnę iš šios aibės pašalinsime tik tada, kai tikrai būsime suradę trumpiausią kelią iki jos.

Q = [Alfa, Beta, Gama, Delta, Omega]

Trumpiausią kelią nuo viršūnės Alfa iki viršūnės i pažymėkime t[i]. Kol kas t[i] ((, jei i ( Alfa. 

Randame visas viršūnes, gretimas pradinei viršūnei Alfa. Kol neradome trumpesnio kelio, atstumus iki šių viršūnių prilyginkime briaunų svoriams (69.2.b pav.):

69.2.b pav.

Iki viršūnės Alfa trumpiausią kelią tikrai suradome, todėl ją pašaliname iš aibės Q:

Q = [Beta, Gama, Delta, Omega]

Aibėje Q randame viršūnę u, atstumas iki kurios Alfa yra trumpiausias. Randame jai gretimą viršūnę v. Jai Jei ta viršūnė jau buvo pasiekta (t[v] ( (), tai dabar turime du variantus, kaip ją pasiekti: arba eiti anksčiau rastu keliu (tuomet atstumas iki jos bus lygus t[v]) arba eiti per viršūnę u (tuomet atstumas iki jos bus lygus t[u] + v(u, v)). Iš šių dviejų atstumų pasirenkame trumpesnį. Taip išnagrinėjame visas viršūnes, gretimas viršūnei u. Viršūnę u pašaliname iš aibės ir vėl ieškome viršūnės, iki kurios jau surastas atstumas nuo Alfa yra trumpiausias. Darbas baigiamas, kai nebelieka nei vienos neišnagrinėtos viršūnės. 

Mūsų pavyzdyje tai trumpiausias atstumas iki viršūnės Alfa yra nuo viršūnės Beta (69.2.b pav.). Viršūnei Beta yra gretimos trys viršūnės: Gama, Delta ir Omega. Apskaičiuojame atstumus iki jų (69.2.c pav.).

Žemiau pateikti paveikslėliai iliustruoja Dijkstros algoritmo veikimą (69.2.a-f pav.):

69.2.a-f pav.

Patį algoritmą užrašysime Pseudo-Paskalio kalba. Mat yra įvairių algoritmo realizacijų, vienos labiau tinka vienokiems, kitos – kitokiems uždaviniams spręsti. Apibendrintas algoritmas tiks visoms situacijoms.



Algoritmo pradiniai duomenys:

1) grafas G, turintis N viršūnių;

2) svorio funkcija s[i, j]  (1( i, j ( N)  



                  0, jei nėra briaunos, jungiančios viršūnę i su viršūne j;

s[i, j] =  

                    briaunos, jungiančios viršūnę i su viršūne j ilgiui, jei 

                    tokia briauna egzistuoja;



3) viršūnė Alfa.



Algoritmo rezultatai:

1) t[i]  (1 ( i ( N) – lentelė, rodanti trumpiausio kelio nuo viršūnės Alfa iki

                               viršūnės i ilgį;

2) k[i] – (1 ( i ( N) – lentelė, padedanti atsekti trumpiausią kelią.



procedure Dijkstra (G, s, alfa; var k, t);

begin

  for i := 1 to N do 

    begin

      t[i] := (;

      k[i] := 0

    end;

  t[alfa] := 0;

  Q := [1..N]; {Q – nenagrinėtų viršūnių aibė}

  while Q ( [] do

    begin

      u := min(Q);   { kreipinys į funkciją, išrenkančią iš aibės Q }

                     { viršūnę su mažiausiu t[i] }

      Q := Q - [u];

      while neperžiūrėtos visos viršūnės v, sujungtos su viršūne u   do

         if t[v] > t[u] + s(u, v)

            then begin

                   t[v] := t[u] + s(u, v);

                   k[v] := u;

                 end

end; { Dijkstra }



O dabar Dijkstros algoritmą pritaikysime mūsų uždaviniui. Akivaizdu, kad trumpiausias kelias tarp taškų (0; 0), (100; 100) yra laužtė, sudaryta iš tam tikro skaičiaus kvadratėlių nekertančių atkarpų. Atkarpų galai gali būti tik kvadratėlių viršutinis kairysis arba apatinis dešinysis kampai bei taškai (0; 0) ir (100; 100). Kadangi visi kvadratėliai yra atskirti vienas nuo kito (t. y. visuomet egzistuoja tarpas tarp jų, kuriuo galima praeiti), nelogiška būtų, jei trumpiausias kelias eitų per kurį nors iš likusių dviejų kvadratėlio kampų.

Taigi sudarome grafą, kurio lankai yra visos galimos atkarpos, jungiančios visų kvadratėlių minėtus kampus, taip pat pradinį ir galinį taškus. Jei lauke yra n kvadratėlių, tai grafas turi 2 ( n + 2 viršūnes. Jas patogu numeruoti nuo 0 iki 2 ( n + 1, kur nulinė ir �(2n+1)-oji viršūnės atitinka taškus (0, 0) ir (100, 100) o (2i-1)-oji viršūnė bus i-ojo kvadratėlio viršutinis kairysis kampas, (2i)-oji viršūnė – to paties kvadratėlio apatinis dešinysis kampas. Šių viršūnių koordinates apskaičiuoti nesunku. Jei kvadratėlio apatinio kairiojo kampo koordinatės yra (ax, bx), tai viršutinio kairiojo kampo koordinatės bus (ax, bx + 5), apatinio dešiniojo – (ax +5, bx).

Grafas turėtų būti orientuotas. Mat į trumpiausią kelią neįeis jokia atkarpa, kuria paėjus sumažėtų nors viena iš koordinačių, t. y. būtų paeita atgal x arba y ašimi. Šį teiginį nesunkiai galima įrodyti matematiškai, tačiau paliekame tai padaryti skaitytojui. 

Sudarant grafą reikėtų atkreipti dėmesį į tai, kad atkarpa, jungiančia du kvadratėlio kampus, galima eiti tik tada, jei ji nekerta kitų kvadratėlių. Kaip tai patikrinti?

Tarkime, kad turime du taškus a ir b bei kvadratėlį k. Kadangi abi taško b koordinatės yra didesnės už taško a koordinates, užtenka paimti kvadratėlio įstrižainę, jungiančią viršutinį kairįjį ir apatinį dešinįjį kampus ir patikrinti, ar ji kerta atkarpą [a; b]. 

Išveskime tiesę per taškus A ir B. Visi taškai (x, y), priklausantys tiesei, tenkins lygtį:

y - A.y        (        x - A.x

B.y - A.y            B.x - A.x

Taškai (x, y) esantys vienoje tiesės pusėje, tenkins nelygybę: 

y - A.y        <        x - A.x

B.y - A.y            B.x - A.x

kitoje tiesės pusėje –

y - A.y        >        x - A.x

B.y - A.y            B.x - A.x

Į šią lygtį įstačius abiejų kvadratėlio k kampų koordinates galima patikrinti, ar abu kampai yra toje pačioje, ar priešingose tiesės pusėse. Jei abu kampai yra toje pačioje pusėje, arba nors vienas jų liečia tiesę, kvadratėlis tikrai nekertamas. Jei kampai yra priešingose pusėse, įstrižainė tikrai kerta atkarpą, bet nebūtinai tiesę (69.3 pav.):

69.3 pav.

Atkarpa [a, b] kirs kvadratėlį k tik tada, kai tiesė išvesta per taškus a ir b kirs kvadratėlio įstrižainę, o tiesė, išvesta per kvadratėlio įstrižainę, kirs atkarpą [a, b].

Turėdami grafą, tiesiogiai pritaikome Dijkstros algoritmą. Programuojant iškils problema: kaip pasielgti su tomis atkarpomis, kurios kerta kvadratėlius? Ar reikia jas iš viso įtraukti į grafą? Jei tokių atkarpų neįtrauksime, tai grafas gali būti nejungus ir reikės rašyti papildomą procedūrą, tikrinančią grafo jungumą. Jei į grafą įtraukiame visas atkarpas, tuomet grafas tikrai bus jungus. Šių atkarpų ilgius prilyginkime kokiam nors dideliam skaičiui ir tuo užtikrinsime, kad tais lankais tikrai nebus einama. Galima susitarti, kad tokių lankų ilgius prilyginsime 200. Tai yra ribinis trumpiausio kelio ilgis (jei nuo pradinio taško į galinį eitume tik horizontaliai ar vertikaliai, ilgis būtų lygus 200). Iš uždavinio sąlygos apribojimų akivaizdu, kad trumpiausio kelio ilgis tikrai bus mažesnis.

Kadangi šiame uždavinyje duomenų įvedimas, jų transformavimas į grafą, rezultatų išvedimas nėra trivialūs, pateikiame ne tik svarbiausias procedūras, bet ir visą programą:

program trumpiausias_kelias;

  const maxn = 30; { maksimalus kvadratėlių skaičius }

  type grafas = array [0..maxn*2+1, 0..maxn*2+1] of real;

                { į grafą reikia įtraukti ir tašką (0, 0) bei (100, 100) }

       koordinatės = array [0..maxn*2+1] of

                         record { visų reikalingų grafo viršūnių koordinatės }

                            x, y: integer

                         end;

       atstumai = array [0..2*maxn+1] of real;

       kelias = array [0..2*maxn+1] of integer;

       aibė = set of 0..2*maxn+1;



  { žemiau išvardyti globalieji kintamieji }

  var n: integer; { kvadratėlių skaičius }

      g: grafas;

      koord: koordinatės; { grafo viršūnių koordinatės }

      pi: kelias;



  procedure įvedimas;

  { įveda pradinius duomenis – kvadratėlių skaičių n }

  { ir kvadratėlių koordinates. Pastarąsias iš karto }

  { perskaičiuoja į grafo viršūnių koordinates }

  { rezultatai – globalieji kintamieji n ir koord }

    var f: text;

        i, a, b: integer;

  begin

    assign(f, 'KELIAS.DAT'); reset(f);

    readln(f, n);

    for i := 1 to n do

      begin

         readln(f, a, b);

         { apatinio dešiniojo kampo koordinatės }

         koord[2*i-1].x := a;

         koord[2*i- ].y := b + 5;

         { viršutinio kairiojo kampo koordinatės }

         koord[2*i].x := a + 5;

         koord[2*i].y := b

      end;

    { taškų (0, 0) ir (100, 100) koordinatės }

    koord[0].x := 0;

    koord[0].y := 0;

    koord[2*n+1].x := 100;

    koord[2*n+1].y := 100

  end; { įvedimas }



  function tiesė (a, b, c: integer): integer;

  { tikrina, kurioje tiesės, einančios per taškus a ir b pusėje yra taškas c }

  { pradinis duomuo – globalusis kintamasis koord }

    var k, d: integer;

  begin

    k := (koord[c].x - koord[a].x) * 

         (koord[b].y - koord[a].y);

    d := (koord[c].y - koord[a].y) * 

         (koord[b].x - koord[a].x);

    if k = d 

       then tiesė := 0 { taškas kerta tiesę }

       else if k < d then tiesė := 1

                     else tiesė := -1

  end; { tiesė }



  procedure grafo_sudarymas;

  { pagal turimas grafo viršūnių koordinates sudaro grafą – }

  { t. y. suskaičiuoja atstumus tarp visų grafo viršūnių }

  { pradiniai duomenys – n ir koord }

  { rezultatas – grafas g }

    var i, j, k: integer;

        kerta: boolean; { ar atkarpa kerta kurį nors kvadratėlį }

  begin

    for i := 0 to 2*n+1 do

      for j := 0 to 2*n+1 do

        begin

          if (koord[i].x > koord[j].x) or 

             (koord[i].y > koord[j].y) or

             (i = j){ atgal neisime, taigi ir atstumo neskaičiuosime }

          then g[i, j] := 200 { fiktyvus atstumas }

          else begin { reikia patikrinti, ar dvi viršūnes jungianti atkarpa }

                     { nekerta kitų kvadratėlių }

                 kerta := false;

                 for k := 1 to n do

                   if (abs(tiesė(i, j, 2*k-1) - 

                           tiesė(i, j, 2*k)) = 2) and

                      (abs(tiesė(2*k-1, 2*k, i) - 

                           tiesė(2*k-1, 2*k, j)) = 2)

                   then begin

                          g[i, j] := 200; { fiktyvus atstumas }

                          kerta := true

                        end;

                   { jei viskas gerai, skaičiuojame atstumą }

                   if not kerta

                   then g[i, j] := 

                           sqrt(sqr(koord[i].x - koord[j].x) +

                                sqr(koord[i].y - koord[j].y))

               end { else }

        end

  end; { grafo_sudarymas }



  procedure Dijkstra;

  { randa trumpiausią kelią grafe g nuo nulinės iki visų likusių viršūnių }

  { pradiniai duomenys – n ir g, rezultatas – pi }

     var i, u, v: integer;

         min: real;

         Q: aibė;

         d: atstumai;

  begin

    for i := 0 to 2*n+1 do

       begin

         d[i] := maxint;

         pi[i] := 0

       end;

    d[0] := 0;

    Q := [0..2*n+1];

    while Q <> [] do

      begin

        { rasime viršūnę, iki kurios atstumas yra trumpiausias }

        min := maxint;

        for i := 0 to 2*n+1 do

          if (d[i] < min) and (i in Q)

          then begin

                   u := i;

                   min := d[i]

                 end;

        Q := Q - [u];

        for v := 0 to 2*n+1 do

          if d[v] > d[u] + g[u, v]

          then begin

                    d[v] := d[u] + g[u, v];

                    pi[v] := u

                  end

     end

  end; { Dijkstra }



  procedure spausdinti;

  { pradiniai duomenys – n ir pi }

    var f: text;

    procedure spausd (v: integer);

    begin

       if v <> 0

          then spausd (pi[v]);

       writeln(f, koord[v].x, ' ', koord[v].y)

    end;

  begin

    assign(f, 'KELIAS.REZ'); rewrite(f);

    spausd(2*n+1);

    close(f)

  end; { spausdinti }



begin

  įvedimas;

  grafo_sudarymas;

  Dijkstra;

  spausdinti

end.

70. Rėmeliai. Šis uždavinys Šeštojoje nacionalinėje informatikos olimpiadoje buvo pripažintas pačiu įdomiausiu uždaviniu.

Sąlygoje pasakyta, kad visų rėmelių matyti bent po vieną kiekvienos kraštinės simbolį. Tai reiškia, kad nesunkiai galime nustatyti kiekvieno rėmelio koordinates. Laikysime, kad duotame paveikslėlyje, sudarytame iš raidžių ir taškų, simboliai numeruojami iš apačios į viršų ir iš kairės į dešinę. O kiekvieną rėmelį nusako jo kairiojo apatinio ir dešiniojo viršutinio kampo koordinatės. 

Taigi pirmiausia, peržiūrime paveikslėlį ir įsidėmime visus skirtingus simbolius (t. y. lotyniškas didžiąsias raides), iš kurių sudaryti rėmeliai, bei nustatome kiekvieno rėmelio koordinates. O tada iš eilės peržiūrime visus rėmelius, ieškodami ant viršaus gulinčio rėmelio. Radę tokį rėmelį, visas jį sudarančias raides pakeičiame taškais. 

Rėmelis guli ant viršaus, jei visi simboliai, esantys rėmelio kraštinėse, yra arba raidė,sudaranti tą rėmelį, arba taškas „.“. Panagrinėkime sąlygoje pateiktą pavyzdį (70.2 pav.):

70.2 pav.

Vienintelis rėmelis, kurio kraštinės sudarytos tik iš tą rėmelį sudarančių raidžių – tai rėmelis C. Nuimame jį, t. y. jį sudarančias raides pakeičiame taškais (70.3 pav.):

70.3 pav.

Dabar matome, kad į rėmelio B kraštines nėra įsiterpusių jokių kitus rėmelius sudarančių raidžių. Nuimame ir jį (70.4 pav.):

70.4 pav.

Dabar jokios pašalinės raidės neįsiterpia į rėmelio A kraštines. Analogiškai kartojame šiuos veiksmus.

Pateikiame šį uždavinį sprendžianti algoritmą:

const max_d = 30; { maksimalios piešinio dimensijos }

type piešinys = array [1..max_d, 1..max_d] of char;

     rėmel = array ['A'..'Z'] 

               of record

                    x1, y1, { rėmelio apatinio kairiojo ir viršutinio dešiniojo }

                    x2, y2: integer; { kampų koordinatės }

                  end;

     aibė = set of char; { aibė skirtingų raidžių, iš kurių sudaryti rėmeliai }



var pieš: piešinys; { sudėliotų rėmelių piešinys }

    a, p: integer;  { piešinio aukštis ir plotis }

    { visi kintamieji – globalieji visoms procedūroms }



  procedure nuimti (rėmaib: aibė; rėm: rėmel;

                    var raidė: char);

                    { raidė, iš kurios sudarytas nuimtasis rėmelis }

  { nuima viršutinį rėmelį, vietoje jo sudeda taško simbolius }

    var rasta: boolean;

        r: char;

        i: integer;

  begin

    { rasime rėmelį, kurį reikia nuimti }

    rasta := false;

    r := 'A';

    while not rasta do  { tikrinsime, ar negalime nuimti rėmelio, }

      if r in rėmaib    { sudaryto iš raidžių r }

      then begin

             rasta := true;

             { tikriname vertikalias rėmelio sienas }

             for i := rėm[r].x1 to rėm[r].x2 do

               begin

                 rasta := (rasta and

                          (pieš[i, rėm[r].y1] in [r, '.']))

                       and(pieš[i, rėm[r].y2] in [r, '.'])

               end;

             { tikriname horizontalias rėmelio kraštines }

             for i := rėm[r].y1 to rėm[r].y2 do

               rasta := rasta and

                     ((pieš[rėm[r].x1, i] in [r, '.']) and

                      (pieš[rėm[r].x2, i] in [r, '.']));

             if not rasta

               then r := succ(r)

           end

      else r := succ(r);

      raidė := r;

      { nuimame raides }

      for i := rėm[r].x1 to rėm[r].x2 do

        begin

          pieš[i, rėm[r].y1] := '.';

          pieš[i, rėm[r].y2] := '.'

        end;

      for i := rėm[r].y1 to rėm[r].y2 do

        begin

          pieš[rėm[r].x1, i] := '.';

          pieš[rėm[r].x2, i] := '.'

        end

  end; { nuimti }



  procedure koordinatės (rėmaib: aibė; var rėm: rėmel);

  { suranda visų rėmelių koordinates – užpildo masyvą rėm }

    var r: char;

        i, j: integer;

  begin

    { inicializuojame masyvą rėm }

    for r := 'A' to 'Z' do

      if r in rėmaib

      then begin

             rėm[r].x1 := max_d;

             rėm[r].y1 := max_d;

             rėm[r].x2 := 0;

             rėm[r].y2 := 0

           end;

    { užpildome jį }

    for i := 1 to a do

      for j := 1 to p do

        if pieš[i, j] <> '.'

        then begin

               r := pieš[i, j];

               { tikriname eilutę }

               if i < rėm[r].x1

                  then rėm[r].x1 := i;

               if i > rėm[r].x2

                  then rėm[r].x2 := i;

               { tikriname stulpelį }

               if j < rėm[r].y1

                  then rėm[r].y1 := j;

               if j > rėm[r].y2

                  then rėm[r].y2 := j

             end

  end; { koordinatės }



procedure ieškoti (var tvarka: string);

{ randama, kokia tvarka buvo sudėlioti rėmeliai }

  var rėmaib: aibė;

      rėm: rėmel;

      i, j: integer;

      raidė: char;

begin

  tvarka := '';

  rėmaib := [];

  { rasime visas raides, iš kurių sudaryti rėmeliai }

  for i := 1 to a do

    for j := 1 to p do

      if pieš[i, j] <> '.'

         then rėmaib := rėmaib + [pieš[i, j]];

  { rasime kiekvieno rėmelio koordinates }

  koordinatės (rėmaib, rėm);

  while rėmaib <> [] do { kol nenuimti visi rėmeliai }

    begin

      nuimti (rėmaib, rėm, raidė);  { nuimamas viršutinis rėmelis }

      tvarka := tvarka + raidė;

      rėmaib := rėmaib - [raidė]

    end

end; { ieškoti }



Testo nr.�Pradiniai duomenys arba jų dydis�Rezultatai��1�6

6

AAAAA.

A.BBAB

A.B.AB

A.BBAB

A...A.

AAAAA.�AB��2�6

13

.......AAAAA.

.EEEBBBBCCCA.

DEDDB.CBAACA.

DE.DB.CBCCC..

DEDDB..B.E...

.EEEBBBBEE...�BCAED��3�7

22

.....CCCC.HHHH..GGGG..

.....C..CFHFEHEEG..G..

.....BBBBFHBEAAAGABG..

.....B.IIFHHHAFIGGGG..

.....BBIBFFFEAFDEADIDD

.....C.IIIIIEAAAAAII.D

.....CCCC..DEEEEEDDDDD�GAHEFIDBC��4�12

28

.JJJJJJJJJJJDDDHHHHHH.......

.J.....D...J...HDBBBBBNNNAAA

CJCLLLLLLLLJLC.HOOOOOONANB.A

CJ.L...D...JLC.HOBMMMONANB.A

CJ.LKKKKKKKIIIIIOIIIIOKKKKKK

CJ.LK..D..GILCGHOBM.HONANB.K

CJJJJJJJJJJILFFHOOOOOONANB.K

CCFLKKKKKKKILKKKKKKKKIKKKKKK

..FLFFFFFFFILFFHFBBBBINNNB.A

...L...D.EGIIIIIIIIIII.A...A

...L...DDEEELDDHHHMHHM.AAAAA

...LLLLLLLLLL.....MMMM......�OIJLKNABMHFCGED��5�24

24�QHMRGPNFKLCTDOAISBEJ��6�30

30�MWLNGOZSDAYUTJPCQXIEHFVBRK��

Šių testų pateikti tik išmatavimai ir rezultatai

